



100%
Markt+Technik

Excel

Das Sparbuch

Finanzen im Griff

J. FLECKENSTEIN B. GEORGI




Markt+Technik



KAPITEL 3

Das Funktionen-Konzentrat



Wenn Sie jemand fragt „Was lässt sich denn mit Excel so alles berechnen?“, werden Sie wahrscheinlich ins Stocken geraten, auch wenn Sie ein versierter Excel-Anwender sind. „Alles Mögliche. Es gibt so viele verschiedene Excel-Funktionen, dass das fast keiner so genau sagen kann. Es gibt zum Beispiel unzählige statistische Funktionen, finanzmathematische Funktionen etc., da muss man schon Mathematiker sein, um da durchzublicken.“ So oder ähnlich könnte die Antwort sein.

3.1 Befreiung vom Dogma der Funktionskategorien

Obwohl die integrierte Excel-Hilfe zu den einzelnen Funktionen teilweise sehr ausführlich und viel besser als ihr Ruf ist, scheuen sich die meisten Anwender, sich wirklich durch den dichten Dschungel des Funktionskatalogs zu kämpfen. Stattdessen werden Berechnungen in VBA programmiert, obwohl für die aktuelle Problemstellung eventuell die passende Excel-Funktion standardmäßig verfügbar ist. Doch woran liegt das?

Ein Grund dafür könnte sein, dass die Kategorisierung der Funktionen nicht besonders hilfreich ist. *Statistische Funktionen*, was heißt das schon? Muss man Statistiker sein, um damit etwas anfangen zu können? Bei manchen Funktionen mag das der Fall sein, aber Funktionen wie MAX oder MIN kann eigentlich jeder Excel-Anwender gebrauchen. Trotz Kategorisierung sind die Funktionen so bunt durcheinandergewürfelt, dass es einfach nicht möglich ist, einen globalen Überblick zu behalten, geschweige denn, sie jemand anderem zu vermitteln. Ein Beispiel:

In der Kategorie *Statistische Funktionen* stehen die Funktionen VARIANZEN, VARIATION und VARIATIONEN unmittelbar untereinander und klingen sehr ähnlich. Funktional haben sie aber überhaupt nichts miteinander zu tun. Die erste ermittelt ein Streuungsmaß, die zweite Werte eines exponentiellen Trends, und die dritte gehört in den Bereich der Kombinatorik. Sie passt damit viel besser zur Funktion KOMBINATIONEN, die sich aber als völliger Außenseiter in der Kategorie *Math. & Trigonometrie* tummelt. Auch der Begriff Matrixfunktionen ist kein aussagekräftiger Begriff für die Funktionen, die sich dahinter verbergen

Um diese Verwirrung zu beseitigen, wollen wir uns von dem Dogma der vorgegebenen Funktionskategorien befreien und Funktionen so gruppieren, wie sie wirklich zusammengehören. Dadurch wird unseres Erachtens ein übergreifendes Verständnis des vollen Funktionsumfangs der Standardfunktionen erheblich erleichtert. Für eine detaillierte Darstellung der kompletten Funktionspalette reicht ein Kapitel natürlich nicht aus. Dies soll nur ein grober Überblick sein und Ihr Interesse für die eine oder

andere Gruppe von Funktionen wecken. Wenn Sie dann tiefer in die Materie einsteigen wollen, geben Sie der integrierten Excel-Hilfe eine Chance, und beherzigen Sie den abgedroschenen, aber dennoch wahren Satz: Probieren geht über Studieren.

Die Funktionen in Excel 2007 wurden insofern aufgewertet, als es nun eine Rubrik innerhalb der Multifunktionsleiste gibt, welche die verschiedenen Kategorien anzeigt (Abbildung 3.1).




Abbildung 3.1: Funktionskategorien werden in Multifunktionsleiste angezeigt.

Abweichend von den vorgegebenen elf Funktionskategorien gliedern wir die Standardfunktionen in folgende 22 Fachgebiete (bei einzelnen Funktionen ist die Zuordnung nicht unumstritten und deshalb letztlich Geschmackssache – sie hätte auch anders erfolgen können). In unserem Erstlingswerk *Excel – Das Zauberbuch* waren es nur 19 Gruppen. Wirklich neue Funktionen gibt es mit SUMMEWENNS, ZÄHLENWENNS, MITTELWERTWENN, MITTELWERTWENNS und WENNFEHLER nur fünf. Diesmal integrieren wir aber die ehemaligen Add-In-Funktionen in den Rundflug. Diese sind in XI2007 in dem Sinne neu, nun zum Standardkatalog zu gehören. Einige davon ließen sich recht gut den vorhandenen 19 Gruppen zuordnen. Für die übrigen wurden drei eigene Pakete geschnürt. Eine davon behandelt ausschließlich Wertpapierfunktionen, ist also besonders für Finanzakrobaten interessant. Die Funktionen der Gruppe *Cube*, die ausschließlich für die Anbindung von OLAP-Datenbanken gedacht sind, behandeln wir nicht.

1.	Datumsfunktionen
2.	Zeitfunktionen
3.	Textfunktionen
4.	Datentypbeschreibung und -umwandlung
5.	Rechnen mit Bedingungen
6.	Rundungs- und Formatierungsfunktionen
7.	Verweisfunktionen
8.	Bereichsrückgabefunktionen
9.	Mathematik allgemein
10.	Lageparameter
11.	Streuungsmaße
12.	Regressionsrechnung
13.	Kombinatorik
14.	Statistische Verteilungen
15.	Trigonometrie
16.	Matrizenrechnung
17.	Zinseszins- und Rentenrechnung
18.	Abschreibungsmethoden
19.	Exoten
20.	Wertpapierfunktionen
21.	Komplexe Zahlen
22.	Umwandlung Zahlensysteme

Zur noch besseren Orientierung ergänzen wir diesmal am Ende des Kapitels eine Gesamtliste aller Funktionen mit Gegenüberstellung von herkömmlicher Funktionskategorie und unserer Zuordnung.

3.2 Datumsfunktionen

Die Funktion **DATUM** ermittelt aus den Bestandteilen *Jahr*, *Monat* und *Tag* ein Datum. In Excel steht hinter jedem Datum eine fortlaufende Ganzzahl, die der Anzahl der Tage entspricht, die zwischen dem 01.01.1900 und diesem Datum vergangen sind. Mit den Funktionen **JAHR**, **MONAT** und **TAG** kann dieses Datum wieder in die Einzelteile Jahr, Monat und Tag zerlegt werden.

Die Funktion **DATWERT** wandelt ein als String angegebenes Datum in die entsprechende fortlaufende Ganzzahl um.

Die Funktion **HEUTE** ermittelt bei jeder Neuberechnung die fortlaufende Ganzzahl des aktuellen Datums.

Die Funktion **WOCHENTAG** ermittelt eine Zahl von 1 bis 7, die den Tagen einer Woche entspricht.

Syntax	Beispiel	Wert	Datum
=DATUM(Jahr;Monat;Tag)	=DATUM(2006;6;9)	38877	09.06.2006
=JAHR(Datum)	=JAHR(38877)	2006	
=MONAT(Datum)	=MONAT(38877)	6	
=TAG(Datum)	=TAG(38877)	9	
=DATWERT(Datumstext)	=DATWERT("09.06.2006")	38877	09.06.2006
=WOCHENTAG(Datum;Typ)	=WOCHENTAG("09.06.2006";2)	5	

Tabelle 3.1: Beispiele zu herkömmlichen Datumsfunktionen

Darüber hinaus gibt es in der Kategorie *Datum* sechs neue Funktionen (ehemals Add-In Funktionen), die aber nicht wirklich Neues bieten, da man sie auch in früheren Versionen durch herkömmliche Funktionen ersetzen konnte. Diese Substitution durch die Veteranen ist zwar nicht mehr wirklich nötig, hilft aber nach wie vor, die Rechenlogik zu verstehen, die dahinter steht.

MONATSENDE liefert den letzten Tag des Monats eines angegebenen Datums, wobei noch eine bestimmte Anzahl Monate voraus oder zurück gerechnet werden kann.

=MONATSENDE(Datum;0)

ergibt dasselbe wie

=DATUM(JAHR(Datum);MONAT(Datum)+1;0)

In Worten gefasst: Der letzte Tag des laufenden Monats ist der „nullte“ Tag des nächsten (+1) Monats. Wird noch ein Tag addiert, hat man stets den ersten Tag des Folgemonats. EDATUM addiert oder subtrahiert ebenfalls zu einem bestimmten Ausgangsdatum eine vorgegebene Anzahl Monate. EDATUM(Datum;Monate) ersetzt die herkömmliche Formel:

=MIN(DATUM(JAHR(A3);MONAT(A3)+B3+{1.0};{0.1}*TAG(A3)))

BRTEILJAHRE wandelt die Anzahl der ganzen Tage zwischen Ausgangsdatum und Enddatum in Bruchteile von Jahren um. Wie bei vielen finanzmathematischen Funktionen kann noch die Jahresbasis eingegeben werden. Bei einer Jahresbasis von 2 mit 360 Tagen gilt die Gleichung:

=BRTEILJAHRE(A1;E1;2)=(E1-A1)/360

Die Funktion NETTOARBEITSTAGE und deren komplizierte Herleitung mit herkömmlichen Funktionen haben wir bereits in Kapitel 3 „Pro-Add-In“ demonstriert. Sie gibt die Anzahl Arbeitstage in einem Zeitintervall zurück. Feiertage in dem Intervall können subtrahiert werden. Welche Feiertage dies sind, muss man aber selbst definieren. ARBEITSTAG macht genau das Gegenteil. Hier wird eine Anzahl von Arbeitstagen vorgegeben, und Excel berechnet das resultierende Enddatum des Intervalls. Feiertage können ebenfalls definiert und berücksichtigt werden.

Schließlich gibt es die Funktion KALENDERWOCHE. Da die Funktion nach amerikanischem Standard rechnet, ist sie für Deutschland unbrauchbar. Wir definieren die erste Kalenderwoche eines Jahres als die Woche, in die mindestens vier Tage fallen (DIN 1355).

Der amerikanische (der weltweit am häufigsten verbreitete) Standard besagt, dass die erste Kalenderwoche eines Jahres die Woche ist, die den 01. Januar enthält. Der Wochentag des Beginns der nächsten Kalenderwochen in den USA ist der Sonntag. Das führt in manchen Jahren zu der kuriosen Situation, dass es an drei aufeinanderfolgenden Tagen drei unterschiedliche Kalenderwochen gibt. Freitag, der 31.12.1999, ist die KW 53, Samstag, der 01.01.2000, die KW 1 und Sonntag, der 02.01.2000, die KW 2. Die korrekte deutsche Berechnung lautet bei Datumsangabe in A1:

=KÜRZEN((-WOCHENTAG(A1;2)-DATUM(JAHR(A1+4-WOCHENTAG(A1;2));1;-10))/7)

3.3 Zeitfunktionen

Die Funktion ZEIT ermittelt aus den Bestandteilen *Stunde*, *Minute* und *Sekunde* eine Uhrzeit. In Excel steht hinter jeder Zeit eine Bruchzahl zwischen 0 und 1. 0,25 steht für 06:00 Uhr morgens, 0,5 steht für 12:00 mittags und 0,75 für 18:00 Uhr abends.

Die Funktionen STUNDE, MINUTE und SEKUNDE zerlegen diese als Bruchzahl ausgedrückte Uhrzeit wieder in ihre Bestandteile Stunde, Minute und Sekunde.

Die Funktion ZEITWERT wandelt eine als Zeichenkette angegebene Uhrzeit in die entsprechende Bruchzahl um.

Die Funktion JETZT ermittelt bei jeder Neuberechnung die fortlaufende Ganzzahl des aktuellen Datums, analog zu HEUTE zuzüglich der Bruchzahl der aktuellen Uhrzeit. Nur die aktuelle Uhrzeit erhalten Sie durch Subtraktion beider Funktionen: =JETZT()-HEUTE().

Syntax	Beispiel	Wert	Datum
=ZEIT(Stunde;Minute;Sekunde)	=ZEIT(15;20;15)	0,6390625	15:20:15
=STUNDE(Zahl)	=STUNDE(0,6390625)		15
=MINUTE(Zahl)	=MINUTE(0,6390625)		20
=SEKUNDE(Zahl)	=SEKUNDE(0,6390625)		15
=ZEITWERT(Zeit)	=ZEITWERT("15:20:15")	0,6390625	15:20:15

Tabelle 3.2: Beispiele zu Zeitfunktionen

Da als Zeichenkette ausgedrückte Zahlen durch Multiplikation mit 1 auch in Zahlen umgewandelt werden können, funktioniert im letzteren Fall auch:

= "15:20:15" * 1 = 0,6390625

3.4 Textfunktionen

Die Funktion CODE rechnet jedes Zeichen in die fortlaufende Zahl des ANSI-Zeichensatzcodes um. ZEICHEN wandelt diesen Code in das entsprechende Zeichen um und ist somit die Umkehrfunktion von CODE. Mit Umkehrfunktion ist gemeint, dass bei Verschachtelung beider Funktionen der Eingabewert der inneren Funktion dem Ergebnis der äußeren Funktion entspricht:

=CODE(ZEICHEN(66))=66

Die eine Funktion macht quasi die Berechnung der anderen Funktion wieder rückgängig.

Die Funktionen **SUCHEN** und **FINDEN** suchen innerhalb eines Textes eine Zeichenfolge und geben die Positionsnummer des Suchtextes innerhalb des durchsuchten Textes zurück. Sie unterscheiden sich voneinander dadurch, dass erstere nicht nach Groß- und Kleinschreibung unterscheidet, letztere schon.

Die Funktion **TEIL** gibt den Teil eines Textes zurück, der sich aus der Vorgabe von Positionsnummer und Länge des Teilstrings ergibt. Sie ist in etwa die Umkehrfunktion von **FINDEN/SUCHEN**.

Wenn die Position des zurückzugebenden Teilstrings links oder rechts beginnt, nehmen Sie statt der Funktion **TEIL** die Funktionen **LINKS** oder **RECHTS**. Häufig ist es sinnvoll, eine dieser Funktionen mit der Funktion **LÄNGE** zu kombinieren, welche die Anzahl Zeichen eines Textes zählt. Mit $=\text{LINKS}(\text{Text}; \text{LÄNGE}(\text{Text}) - 3)$ können Sie beispielsweise die drei ersten Zeichen von rechts abschneiden.

Die Funktion **VERKETTEN** hängt verschiedene Texte aneinander, was allerdings durch das kaufmännische **&** genauso gut erledigt werden kann. **WIEDERHOLEN** verkettet ebenfalls Text. Hierbei ist es aber immer der gleiche Text, und Sie können nur vorgeben, wie oft er wiederholt werden soll.

Die Funktionen **KLEIN** und **GROSS** wandeln alle Zeichen eines Textes in Klein- oder Großbuchstaben um. Auf numerische Ziffern oder Sonderzeichen haben diese Funktionen keine Auswirkung. **GROSS2** schreibt nur das erste Zeichen jedes Wortes in einem Text groß und den Rest klein. Die Wörter müssen nicht unbedingt durch Leerzeichen getrennt sein. Eine Trennung durch Bindestrich, Unterstrich oder ein anderes Sonderzeichen hat den gleichen Effekt.

GLÄTTEN entfernt überflüssige Leerzeichen (mehr als eines nebeneinander) eines Textes und entspricht in etwa den Trimm-Funktionen, die in diversen Programmiersprachen bekannt sind. **SÄUBERN** entfernt alle nicht druckbaren Zeichen.

IDENTISCH prüft, ob zwei Texte identisch sind, und unterscheidet sich von der einfachen Prüfung =Text1=Text2 dadurch, dass sie zwischen Groß- und Kleinschreibung unterscheidet. Für Excel ist standardmäßig ="a"="A"=WAHR . Die Funktion **DELTA** macht dasselbe, funktioniert aber nur für numerische Zeichen, was ihr jegliche Daseinsberechtigung entzieht. Statt **WAHR/FALSCH** liefert sie 1/0.

Die Funktion WECHSELN ersetzt eine Zeichenfolge innerhalb eines Textes durch eine andere. Die Funktion ERSETZEN macht fast das Gleiche. Der Unterschied besteht darin, dass Sie bei WECHSELN eine Zeichenfolge vorgeben, die innerhalb des Textes gesucht werden muss. Bei ERSETZEN geben Sie stattdessen die Position und die Länge der Zeichenfolge vor, die ausgetauscht werden soll.

Text	Beispiel	Ergebnis
χ	=CODE(Text)	88
88	=ZEICHEN(Text)	χ
Excel	=SUCHEN("e";Text)	1
Excel	=FINDEN("e";Text)	4
Excel	=TEIL(Text;4;1)	e
Excel	=LINKS(Text;2)	Ex
Excel	=RECHTS(Text;2)	el
	=VERKETTEN("E";"x";"c";"e";"l")	Excel
	="E"&"x"&"c"&"e"&"l"	Excel
Ha	=WIEDERHOLEN(Text;3)	HaHaHa
Formeln	=GROSS(Text)	FORMELN
Formeln	=KLEIN(Text)	formeln
KLAUS KÜHNLEIN	=GROSS2(Text)	Klaus Kühnlein
Klaus Kühnlein	=GLÄTTEN(Text)	Klaus Kühnlein
a b	=SÄUBERN(Text)	ab
	=IDENTISCH("a";"A")	FALSCH
	="a"="A"	WAHR
	=DELTA("3";"3")	1
	=DELTA("a";"A")	#WERT!
Alphorn	=WECHSELN(Text;"hor";"i")	Alpin
Alphorn	=ERSETZEN(Text;4;3;"i")	Alpin

Tabelle 3.3: Beispiele zu Textfunktionen

3.5 Datentypbeschreibung und -umwandlung

Unter Programmierern gehört es zum elementaren Basiswissen, dass Daten unterschiedliche Datentypen besitzen. Die in Excel relevanten Datentypen sind *Zahl*, *Text* (String), *boolescher Wert* (WAHR/FALSCH) und *Fehlerwert*.

Die Funktion T wandelt ein Argument in einen Text um. Zahlen werden in eine leere Zeichenfolge (Text mit Länge 0) umgewandelt. Die Funktionen N und WERT wandeln ein Argument in eine Zahl um. Da sich ein alphanumerisches Zeichen nicht wirklich in eine Zahl umwandeln lässt (vom Zeichencode einmal abgesehen), liefert WERT in diesem Fall den Fehlerwert #WERT!. N macht stattdessen eine 0 daraus. N eignet sich unter anderem prima dazu, Kommentare in Formeln zu platzieren, ohne Einfluss auf das Ergebnis zu nehmen. Beispiel:

=IKV({-10.4.4.4};0)+N("Diese Formel berechnet den internen Zinsfuß einer Zahlungsreihe")

Die Funktionen TYP und FEHLER.TYP geben eine Ganzzahl zurück, die angibt, um welchen Datentyp bzw. Fehlertyp es sich bei einem Argument handelt.

Eine ähnliche Aufgabe erfüllen die Funktionen ISTFEHL, ISTBEZUG, ISTFEHLER, ISTKTEXT, ISTLEER, ISTLOG, ISTNV, ISTTEXT und ISTZAHL. Das Ergebnis dieser Funktionen ist aber keine Ganzzahl, sondern ein boolescher Wert: WAHR oder FALSCH.

Mit ISTGERADE und ISTUNGERADE können Sie prüfen, ob eine Zahl durch zwei teilbar ist. Sie ignorieren Nachkommastellen und rechnen das Gleiche wie:

ISTGERADE: =REST(KÜRZEN(A1);2)=0

ISTUNGERADE: =REST(KÜRZEN(A1);2)=1

VORZEICHEN gibt an, ob eine Zahl positiv, negativ oder null ist.

Wert/Text	Beispiel	Ergebnis
123	=T(Zellwert)	
A	=WERT(Zellwert)	#WERT!
A	=N(Zellwert)	0
#NV	=TYP(Zellwert)	16
#NV	=FEHLER.TYP(Zellwert)	7

Wert/Text	Beispiel	Ergebnis
#DIV/0!	=ISTFEHL(Zellwert)	WAHR
123	=ISTBEZUG(Zellwert)	WAHR
#NV	=ISTFEHLER(Zellwert)	WAHR
123	=ISTKTEXT(Zellwert)	WAHR
A	=ISTLEER(Zellwert)	FALSCH
WAHR	=ISTLOG(Zellwert)	WAHR
#NV	=ISTNV(Zellwert)	WAHR
123	=ISTTEXT(Zellwert)	FALSCH
123	=ISTZAHL(Zellwert)	WAHR
2,12	=ISTGERADE(Zellwert)	WAHR
123	=VORZEICHEN(Zellwert)	1

Tabelle 3.4: Beispiele zu Datentypsfunctionen

Die Funktion ANZAHLLEEREZELLEN zählt leere Zellen und Zellen mit der Länge 0. Eine Zelle mit der Länge 0 muss nicht unbedingt leer sein, sie könnte auch einen sogenannten Nullstring "=" enthalten. Die Funktion ISTLEER liefert FALSCH, wenn eine Zelle einen solchen Leerstring enthält. Für die Funktion ANZAHLLEEREZELLEN ist sogar eine Zelle mit der Formel

=WENN(1;"")

leer, und sie zählt deshalb eine 1 für diese Zelle. Sehr vielseitig ist die ehemalige Add-In-Funktion UMWANDELN, die eine in einer bestimmten Maßeinheit angegebene Zahl in eine andere Maßeinheit umwandelt. Die Maßeinheiten sind in folgende Kategorien unterteilt:

- Gewichte: z.B. Gramm, Unze
- Entfernung: Meter, Meile, Seemeile, Fuß, Yard, Zoll, Angstrom, Pica
- Zeit: Jahr, Tag, Stunde, Minute, Sekunde
- Druck: Pascal, Atmosphäre; mm Quecksilber
- Kraft: z.B. Newton

- **Energie:** z.B. Joule, Kalorien, Wattstunde
- **Magnetismus:** Tesla, Gauss
- **Temperatur:** Celsius; Fahrenheit, Kelvin
- **Flüssigmaße:** z.B. Esslöffel, Tasse, Liter, Gallone

Metrische Maße wie Gramm und Meter können außerdem in die Einheitenpräfixe *kilo*, *mega*, *giga* usw. umgewandelt werden.

Beispielsweise entspricht eine Seemeile

=UMWANDELN(1;"Nmi";"km") = 1,852

Kilometern. Stammen die Parameter *Von_Maßeinheit* und *In_Maßeinheit* nicht aus der gleichen Gruppe, gibt es einen Fehlerwert. Logo, denn sieben Esslöffel lassen sich nicht in Grad Fahrenheit umrechnen.

3.6 Rechnen mit Bedingungen

Ein Gebiet, das auch allen Programmierern und Datenbankentwicklern sofort ein Begriff ist, ist das Rechnen mit Bedingungen. Mit zentraler Bedeutung fallen einem dazu die Begriffe WENN – UND – ODER – DANN – SONST ein.

Die Funktionen UND, ODER und NICHT liefern Wahrheitswerte, die zum Ausdruck bringen, ob eine Bedingung erfüllt ist oder nicht. Die Wahrheitswerte WAHR und FALSCH besitzen gleichnamige Funktionen, die völlig sinnfrei sind und laut Excel-Hilfe nur aus Kompatibilitätsgründen zu anderen Programmen zur Verfügung stehen. =WAHR() liefert WAHR. Das ist so, als gäbe es eine Funktion =EINS(), die den Wert 1 liefert.

Die Funktion WENN kann abhängig von einer solchen Bedingungsprüfung eine *Dann*-Berechnung beziehungsweise eine *Sonst*-Berechnung durchführen.

Die Funktion WAHL kann – im Gegensatz zu den zwei Alternativen (*Dann_Wert*/*Sonst_Wert*) der WENN-Funktion – 254 alternative Berechnungen durchführen. Die Prüfung erfolgt deshalb nicht über die Wahrheitswerte WAHR und FALSCH, sondern über einen fortlaufenden Index von 1 bis 254. Vor Excel 2007 konnte man nur 29 Alternativen eingeben.

Die Funktionen WENN und WAHL überprüfen (standardmäßig) eine Zelle auf eine oder mehrere Bedingungen. Die Funktionen ZÄHLENWENN und SUMMEWENN fragen einen ganzen Bereich ab, können dabei aber nur eine Bedingung prüfen. ZÄHLENWENN zählt alle Zellen, auf welche die Bedingung zutrifft; SUMMEWENN summiert für jede Zelle mit erfüllter Bedingung den Zellwert dieser oder benachbarter bzw. gleich indizierter Zellen.

Die Datenbankfunktionen DBSUMME, DBANZAHL, DBMAX, DBMIN usw. können beliebig viele Zellen (Datensätze) hinsichtlich beliebig vieler Bedingungen überprüfen. Damit vereinen sie die Möglichkeiten von WENN, SUMMEWENN und ZÄHLENWENN und sind damit sehr mächtig. Außerdem können sie nicht nur zählen und summieren, sondern auch weitere statistische Größen, beispielsweise Standardabweichung und Varianz, berechnen. Leider ist ihre Handhabung etwas umständlich, und sie arbeiten nicht autark, da jede unterschiedliche Berechnung einen Bereich in der Tabelle verlangt, in dem die Suchkriterien definiert werden.

Aufgrund dieser umständlichen Bedienung sind die Datenbankfunktionen nicht besonders beliebt. Trotzdem besteht eine rege Nachfrage nach Auswertungen mit mehreren Bedingungen. Deshalb wurden in Excel 2007 die beiden neuen Funktionen SUMMEWENNS und ZÄHLENWENNS geschaffen, die mehrere Bedingungen auf einmal verarbeiten können. Und für die goldene Mitte gibt es mit MITTELWERTWENN und MITTELWERTWENNS nun auch Entsprechungen. Mit MINWENN und MAXWENN hätte man das konsequent fortführen können, hat man aber nicht. Schade, darauf müssen wir wohl bis XL???? warten.

Formel	Ergebnis
=WENN(WAHR;3;5)	3
=ODER(WAHR;FALSCH)	WAHR
=UND(WAHR;FALSCH)	FALSCH
=UND(WAHR;WAHR)	WAHR
=WENN(UND(ODER(1;0);1);UND(0;1);5)	FALSCH
=UND(ODER(1;1);NICHT(FALSCH))	WAHR
=WAHL(2;"eins";"zwei";"drei")	zwei

G16							f _x	=MITTELWERTWENNS(C:C;A:A;E7;B:B;F7)
	A	B	C	D	E	F	G	
1	Kriterium1	Kriterium2	Wert		Formel		Ergebnis	
2	b	x	3		=ZÄHLENWENN(A:A;"a")		6	
3	a	x	6		=SUMMEWENN(A:A;">a";C:C)		45	
4	a	x	4		=MITTELWERTWENN(A:A;"a";C:C)		5,166667	
5	b	y	4					
6	c	z	2		Kriterium1	Kriterium2		
7	a	z	7		a	y		
8	a	y	6		b	x		
9	b	y	4					
10	c	y	6		=DBSUMME(A:C;"Wert";E6:F7)		13	
11	b	x	7		=DBMAX(A:C;"Wert";E6:F8)		7	
12	c	x	10		=DBANZAHL2(A:A;"Kriterium1";E6:E8)		10	
13	a	y	7					
14	c	y	9		=SUMMEWENNS(C:C;A:A;E7;B:B;F7)		13	
15	a	x	1		=ZÄHLENWENNS(A:A;E7;B:B;F7)		2	
16					=MITTELWERTWENNS(C:C;A:A;E7;B:B;F7)		6,5	
17								

Abbildung 3.2: Rechnen mit Bedingungen

Eine geniale Erfindung ist die längst überfällige Funktion WENNFEHLER, mit der ein wesentlich vereinfachter Fehlerabfang in Formeln möglich wird; zum Beispiel beim SVRWEIS. Wenn diese Funktion einen Suchbegriff nicht finden kann, liefert sie den Fehlerwert #NV. Stattdessen möchte man eventuell lieber den Text „nicht vorhanden“ sehen. Bisher hätte man in diesem Fall wie folgt formuliert:

=WENN(ISTNV(SVERWEIS("x";A:C;3;0));"nicht vorhanden"; SVERWEIS("x";A:C;3;0))

Zum einen nervt die doppelte Eingabe desselben Formelteils. Zum anderen bindet die redundante Berechnung natürlich unnötige CPU-Ressourcen. Mit der neuen Funktion genügt stattdessen:

=WENNFEHLER(SVERWEIS("x";A:C;3;0); "nicht vorhanden")

Der redundante Formelteil wird nicht mehr benötigt, Excel weiß auch so, was zu tun ist.

3.7 Rundungs- und Formatierungsfunktionen

Die Funktion TEXT formatiert eine Zahl und wandelt das Ergebnis in einen Text um. Fast alle Zahlenformate, die sich über den Dialog *Zellen formatieren* einstellen lassen, können Sie ebenso mit dieser Funktion erzeugen. Somit können Sie mit dieser Funktion unter anderem auch runden. Excel bietet eine ganze Reihe weiterer Rundungsfunktionen, die auf den ersten Blick oft gleiche Ergebnisse liefern, aber trotzdem feine Unterschiede haben, die man nicht außer Acht lassen darf.

Die Funktionen RUNDEN, FEST und DM runden nach einer beliebigen Anzahl Stellen auf oder ab. Ab 0,5 wird stets aufgerundet. Sie liefern das identische Ergebnis, außer dass FEST und DM Texte liefern. DM liefert zusätzlich das gemäß Ländereinstellung aktuell gültige Währungssymbol. Bei DM setzt die Multiplikation mit 1 voraus, dass das Währungssymbol mit einer Textfunktion abgeschnitten wurde. Die nicht dokumentierte Funktion USDOLLAR macht exakt dasselbe.

Die Funktionen ABRUNDEN, KÜRZEN, UNTERGRENZE und GANZZAHL runden nach unten ab. Bei ABRUNDEN und KÜRZEN kann eine beliebige Anzahl Stellen vorgegeben werden. Enthält das Argument *Anzahl_Stellen* eine negative Zahl, kann auf volle Zehnerpotenzen (vor dem Komma) gerundet werden. Mit UNTERGRENZE kann man noch feiner justieren, da auf das nächstliegende Vielfache einer beliebig definierbaren Schrittweite abgerundet werden kann. GANZZAHL schneidet einfach alle Nachkommastellen ab. Die gleiche Funktion mit zwei G, also GGANZZAHL führt die völlig banale Prüfung

$= (A1 > B1) * 1$

durch. QUOTIENT ist keinen Deut nützlicher und liefert lediglich den ganzzahligen Teil einer Division, genau wie:

GANZZAHL(Zahl1/Zahl2)

AUFRUNDEN und OBERGRENZE runden nach oben auf und funktionieren ansonsten wie ihre Pendanten beim Abrunden.

VRUNDEN rundet auf das beliebige Vielfache einer Zahl auf oder ab und macht damit das Gleiche wie:

$= \text{RUNDEN}(\text{Zahl} / \text{Vielfaches}; 0) * \text{Vielfaches}$

VRUNDEN liefert stets entweder das Ergebnis von UNTERGRENZE (falls abgerundet wird) oder das von OBERGRENZE (falls aufgerundet wird).

GERADE und UNGERADE runden positive Zahlen auf die nächste gerade bzw. ungerade Zahl auf und im negativen Bereich ab.

ABS liefert den Absolutwert einer Zahl.

Wert	Formel	Ergebnis
2,336	=TEXT(Wert;"#.##0,00")	2,34
2,336	=DM(Wert;2)	2,34 €
2,336	=FEST(Wert;2)	2,34

Wert	Formel	Ergebnis
2,336	=ABRUNDEN(Wert;2)	2,33
2,336	=ABS(Wert)	2,336
2,336	=AUFRUNDEN(Wert;2)	2,34
2,336	=GANZZAHL(Wert)	2
2,336	=GERADE(Wert)	4
2,336	=KÜRZEN(Wert;2)	2,33
2,336	=OBERGRENZE(Wert;0,05)	2,35
2,336	=VRUNDEN(Wert;0,05)	2,35
2,336	=RUNDEN(Wert;2)	2,34
2,336	=UNGERADE(Wert)	3
2,336	=UNTERGRENZE(Wert;0,05)	2,3
12345	=RUNDEN(Wert;-3)	12000

Tabelle 3.5: Beispiele zu Rundungsfunktionen

3.8 Verweisfunktionen

Schauen Sie noch einmal auf die angebotenen Funktionskategorien auf die erste Abbildung dieses Kapitels. Fällt Ihnen etwas auf? Excel spricht jetzt auch von einer Gruppe Verweisfunktionen („*Nachschlagen und Verweisen*“) und gibt damit indirekt zu, dass der Name der Kategorie *Matrixfunktionen* irreführend ist. Hat sich Microsoft das etwa von uns abgeschaut? Immerhin haben wir diese Funktionsgruppe auch in *Excel – Das Zauberbuch* schon so benannt ;-).

Verweisfunktionen durchsuchen Bereiche nach Suchkriterien. Wer mit großen Datenmengen hantiert und verschiedene Datentabellen miteinander in Beziehung setzen muss, benötigt Verweise. In einer Datenbank geschieht dies durch die Verknüpfung von Datenbanktabellen über (in der Regel) sogenannte 1:n-Beziehungen. In Excel wird diese Methodik mit Verweisfunktionen durchgeführt.

SVERWEIS(Suchkriterium;Matrix;Spaltenindex;Bereich_Verweis) durchsucht die linke Spalte eines Bereichs oder einer Matrix nach einem Suchkriterium und gibt vom ersten Treffer eine rechts benachbarte Zelle zurück. WVERWEIS macht das Gleiche, nur 90 Grad versetzt, durchsucht also die oberste Zeile eines Bereichs oder einer Matrix und gibt einen Wert zurück, der sich in derselben Spalte wie der Treffer befindet. Bei beiden Funktionen kann vorgegeben werden, ob nur dann ein Ergebnis geliefert werden soll, wenn eine genaue Übereinstimmung mit dem Suchkriterium besteht. Soll auch ein Ergebnis geliefert werden, wenn es keine genaue Übereinstimmung gibt, muss die Suchspalte oder -zeile aufsteigend sortiert sein. Dann wählt die Funktion den größten Treffer aus, der kleiner oder gleich dem Suchkriterium ist. Diese Variante ist dann sinnvoll, wenn innerhalb von Intervallen gesucht werden soll. Klassischer Anwendungsfall ist hierbei die umsatzabhängige Provisionsvergütung oder die Schulnote, die von der erreichten Punktzahl abhängig ist.

VERWEIS(Suchkriterium;Suchvektor;Ergebnisvektor) ohne S und W erfüllt fast den gleichen Zweck und ist dabei hinsichtlich der Suchrichtung etwas flexibler. Sie kann zeilen- oder spaltenweise suchen, je nach Vorgabe. Sie könnte sogar in einer Zeile suchen, das Ergebnis aber aus einer Spalte liefern. Andererseits muss bei VERWEIS der Suchvektor stets aufsteigend sortiert sein, um ein verlässliches Ergebnis zu gewährleisten. Geliefert wird immer der bestmögliche Treffer, der kleiner oder gleich dem Suchkriterium ist. Da in der Mehrzahl der Fälle allerdings nach einer genauen Übereinstimmung gesucht wird, sind die Einsatzgebiete von VERWEIS begrenzt.

VERGLEICH(Suchkriterium;Suchmatrix;Vergleichstyp) durchsucht ebenfalls wahlweise eine Spalte (vertikale Matrix) oder eine Zeile (horizontale Matrix) und sucht entweder nach einer genauen Übereinstimmung oder dem nächstgelegenen Treffer. Zusätzlich zu (W/S)VERWEIS kann der Suchvektor hier auch absteigend sortiert sein, dann wird der kleinste Eintrag gewählt, der größer oder gleich dem Suchkriterium ist. In diesem Fall muss die Suchmatrix absteigend sortiert sein. Bei dieser Funktion muss keine Ergebniszeile oder -spalte angegeben werden, da sie als Ergebnis eine Ganzzahl liefert, welche die Position des Suchkriteriums innerhalb der zu durchsuchenden Zeile (horizontale Matrix) oder Spalte (vertikale Matrix) darstellt.

Die Datenbankfunktion DBAUSZUG(Datenbank;Datenbankfeld;Suchkriterien) sucht immer vertikal und kann nach links oder rechts blicken. Sie ist die einzige Verweisfunktion, die standardmäßig mehrere Bedingungen verarbeiten kann. Sie sucht stets nach einer genauen Übereinstimmung. Wenn die Suchkriterien aber auf mehrere Datensätze zutreffen, nimmt sie nicht den ersten Treffer (wie die anderen Verweisfunktionen), sondern liefert eine Fehlermeldung.

	A	B	C	D	E	F	G
1	Kriterium1	Kriterium2	Wert		Formel		Ergebnis
2	m	10	96,23		=SVERWEIS("c";A:C;3;0)		64,99
3	g	12	88,6		=SVERWEIS(15;B:C;2;1)		24,1
4	d	14	24,1		=SVERWEIS(15;B:C;2;0)		#NV
5	l	16	29,37		=WVERWEIS("Kriterium2";A1:C15;5;0)		16
6	a	18	55,8		=VERGLEICH("a";A:A;0)		6
7	c	20	64,99		=VERGLEICH("a";A2:A10;0)		5
8	i	22	13,66		=VERGLEICH(35;B:B;1)		14
9	h	24	8,67				
10	b	26	21,63		Kriterium1	Kriterium2	
11	k	28	74,06		h	24	
12	e	30	41,2				
13	f	32	78,66		=DBAUSZUG(A:C;"Wert";E10:F11)		8,67
14	j	34	57,2				
15	n	36	69,38				

Abbildung 3.3: Verweisfunktionen

3.9 Bereichsrückgabefunktionen

Die Funktionen dieser Gruppe sind 100 % Excel-spezifisch. Zeilen, Spalten und Zellen beschreiben Excel-Bereiche bzw. -Tabellen, die sozusagen das lebensnotwendige Skelett einer jeden Excel-Anwendung darstellen.

Mit den Funktionen **BEREICH.VERSCHIEBEN**(Bezug;Zeilen;Spalten; Höhe;Breite), **INDEX**(Bezug;Zeile;Spalte) und **INDIREKT**(Bezug) werden Excel-Bereiche definiert und verändert. Die ersten beiden können aus einem vorgegebenen Ausgangsbereich eine ganze Spalte, Zeile oder einzelne Zelle zur weiteren Verarbeitung herauspicken. Nur bei **INDEX** kann statt des Bereichs auch eine Matrix(konstante) vorgegeben werden. Mit **BEREICH.VERSCHIEBEN** können Bereiche sogar völlig neu dimensioniert und verschoben werden. **INDIREKT** wandelt eine Bereichsangabe im Textformat in einen Bereich um. Dabei kann der Bereich (Bezug) direkt als Text oder aber auch als Bezug auf eine Zelle, die den Bereich in Textform enthält, angegeben werden. Die Bereichsangabe kann auch auf eine nicht aktive Tabelle und sogar auf eine andere Arbeitsmappe verweisen, die aber geöffnet sein muss, um ein Ergebnis zu erhalten.

MTRANS transponiert einen Bereich oder eine Matrix, macht also aus Spalten Zeilen und aus Zeilen Spalten.

Die Funktionen ZEILE und SPALTE liefern die fortlaufende Ganzzahl der Zeile bzw. Spalte des angegebenen Bezugs. Enthält der angegebene Bezug mehrere Zeilen oder Spalten, lassen sich flexibel einsetzbare Zahlenfolgen erzeugen, worauf im Laufe dieses Buches noch sehr ausführlich eingegangen wird. ZEILEN und SPALTEN geben die Anzahl der Zeilen und Spalten des angegebenen Bereichs zurück.

ADRESSE gibt aus der Angabe von Zeilen- und Spaltennummer eine Zelladresse in Textformat aus. Da INDIREKT eine solche Zelladresse in Textform erwartet, können diese beiden Funktionen gut miteinander kombiniert werden.

Die Funktion BEREICHE zählt die innerhalb eines Bezugs aufgeführten Bereiche.

	A	B	C	D	E	F	G	H
1	A1	B1	C1	D1	E1		Formel	Bereich
2	A2	B2	C2	D2	E2			
3	A3	B3	C3	D3	E3		=BEREICH.VERSCHIEBEN(A1:A2;4;1;3;2)	B5:C7
4	A4	B4	C4	D4	E4			
5	A5	B5	C5	D5	E5		=INDEX(A1:E10;9;)	A9:E9
6	A6	B6	C6	D6	E6			
7	A7	B7	C7	D7	E7		=INDIREKT("A"&9&"E"&9)	A9:E9
8	A8	B8	C8	D8	E8			
9	A9	B9	C9	D9	E9		=INDEX(A1:E10;2;4)	D2
10	A10	B10	C10	D10	E10			
11								
12		A1	A2					
13		B1	B2					

Abbildung 3.4: Bereichsrückgabefunktionen

Formel	Ergebnis
=ZEILE(A3)	3
=SPALTE(B3)	2
=ZEILEN(A1:E10)	10
=SPALTEN(A1:E10)	5
=BEREICHE((A1:B2;C5:D6))	2
=ADRESSE(20;30)	\$AD\$20